
Notice 
1. All information included in this document is current as of the date this document is issued. Such information, however, is 

subject to change without any prior notice. Before purchasing or using any Renesas Electronics products listed herein, please 
confirm the latest product information with a Renesas Electronics sales office. Also, please pay regular and careful attention to 
additional and different information to be disclosed by Renesas Electronics such as that disclosed through our website. 

2. Renesas Electronics does not assume any liability for infringement of patents, copyrights, or other intellectual property rights 
of third parties by or arising from the use of Renesas Electronics products or technical information described in this document.  
No license, express, implied or otherwise, is granted hereby under any patents, copyrights or other intellectual property rights 
of Renesas Electronics or others. 

3. You should not alter, modify, copy, or otherwise misappropriate any Renesas Electronics product, whether in whole or in part. 
4. Descriptions of circuits, software and other related information in this document are provided only to illustrate the operation of 

semiconductor products and application examples.  You are fully responsible for the incorporation of these circuits, software, 
and information in the design of your equipment.  Renesas Electronics assumes no responsibility for any losses incurred by 
you or third parties arising from the use of these circuits, software, or information. 

5. When exporting the products or technology described in this document, you should comply with the applicable export control 
laws and regulations and follow the procedures required by such laws and regulations.  You should not use Renesas 
Electronics products or the technology described in this document for any purpose relating to military applications or use by 
the military, including but not limited to the development of weapons of mass destruction.  Renesas Electronics products and 
technology may not be used for or incorporated into any products or systems whose manufacture, use, or sale is prohibited 
under any applicable domestic or foreign laws or regulations. 

6. Renesas Electronics has used reasonable care in preparing the information included in this document, but Renesas Electronics 
does not warrant that such information is error free.  Renesas Electronics assumes no liability whatsoever for any damages 
incurred by you resulting from errors in or omissions from the information included herein. 

7. Renesas Electronics products are classified according to the following three quality grades:  “Standard”, “High Quality”, and 
“Specific”.  The recommended applications for each Renesas Electronics product depends on the product’s quality grade, as 
indicated below.  You must check the quality grade of each Renesas Electronics product before using it in a particular 
application.  You may not use any Renesas Electronics product for any application categorized as “Specific” without the prior 
written consent of Renesas Electronics.  Further, you may not use any Renesas Electronics product for any application for 
which it is not intended without the prior written consent of Renesas Electronics.  Renesas Electronics shall not be in any way 
liable for any damages or losses incurred by you or third parties arising from the use of any Renesas Electronics product for an 
application categorized as “Specific” or for which the product is not intended where you have failed to obtain the prior written 
consent of Renesas Electronics.  The quality grade of each Renesas Electronics product is “Standard” unless otherwise 
expressly specified in a Renesas Electronics data sheets or data books, etc. 

“Standard”: Computers; office equipment; communications equipment; test and measurement equipment; audio and visual 
equipment; home electronic appliances; machine tools; personal electronic equipment; and industrial robots. 

“High Quality”: Transportation equipment (automobiles, trains, ships, etc.); traffic control systems; anti-disaster systems; anti-
crime systems; safety equipment; and medical equipment not specifically designed for life support. 

“Specific”:  Aircraft; aerospace equipment; submersible repeaters; nuclear reactor control systems; medical equipment or 
systems for life support (e.g. artificial life support devices or systems), surgical implantations, or healthcare 
intervention (e.g. excision, etc.), and any other applications or purposes that pose a direct threat to human life. 

8. You should use the Renesas Electronics products described in this document within the range specified by Renesas Electronics, 
especially with respect to the maximum rating, operating supply voltage range, movement power voltage range, heat radiation 
characteristics, installation and other product characteristics. Renesas Electronics shall have no liability for malfunctions or 
damages arising out of the use of Renesas Electronics products beyond such specified ranges. 

9. Although Renesas Electronics endeavors to improve the quality and reliability of its products, semiconductor products have 
specific characteristics such as the occurrence of failure at a certain rate and malfunctions under certain use conditions. Further, 
Renesas Electronics products are not subject to radiation resistance design.  Please be sure to implement safety measures to 
guard them against the possibility of physical injury, and injury or damage caused by fire in the event of the failure of a 
Renesas Electronics product, such as safety design for hardware and software including but not limited to redundancy, fire 
control and malfunction prevention, appropriate treatment for aging degradation or any other appropriate measures.  Because 
the evaluation of microcomputer software alone is very difficult, please evaluate the safety of the final products or system 
manufactured by you. 

10. Please contact a Renesas Electronics sales office for details as to environmental matters such as the environmental 
compatibility of each Renesas Electronics product.  Please use Renesas Electronics products in compliance with all applicable 
laws and regulations that regulate the inclusion or use of controlled substances, including without limitation, the EU RoHS 
Directive.  Renesas Electronics assumes no liability for damages or losses occurring as a result of your noncompliance with 
applicable laws and regulations. 

11. This document may not be reproduced or duplicated, in any form, in whole or in part, without prior written consent of Renesas 
Electronics. 

12. Please contact a Renesas Electronics sales office if you have any questions regarding the information contained in this 
document or Renesas Electronics products, or if you have any other inquiries. 

(Note 1) “Renesas Electronics” as used in this document means Renesas Electronics Corporation and also includes its majority-
owned subsidiaries. 

(Note 2) “Renesas Electronics product(s)” means any product developed or manufactured by or for Renesas Electronics. 

1. 740 族的基本程序 
740 族指令的特点概括如下： 

 

（1）能有效使用 ROM 区的高效率的指令群和丰富的寻址方式。 

（2）能对累加器、存储器、I/O 进行处理的位操作指令和位测试/转移指令。 

（3）丰富的中断源和处理功能。 

（4）字节单位处理和表参照功能卓绝的变址寻址功能。 

（5）不需要软件补正的 10 进制运算功能。 

（6）不经累加器就能进行的存储器和存储器之间、I/O 和 I/O 之间、存储器和 I/O 之间的运算功能。 

2. 有效利用 740 族固有指令的处理 

2.1 存储器间的运算 
（1）加法运算（助记符 ADC） 

ADC 指令在 X 变址模式标志 T 为 1 时将 M（X）、存储器 M 和进位标志 C 的内容相加，结果存入 M（X）
和进位标志 C。此时，累加器 A 的内容不变，而状态标志发生变化。其中，M（X）为变址寄存器 X 指向地址
的存储器。 

例） 

CLC ① 

SET ② 

LDX #ADDATA ③ 

ADC $10, X ④ 

CLT ⑤ 

 

（说明） 

①将进位标志 C 的内容置 0。 

②将 X 变址模式标志 T 的内容置 1。 

③将#ADDATA（例：70H）装入变址寄存器 X。 

④将存储器的 70H 地址的内容、80H 地址（70H+10H）的内容和进位标志 C 的内容相加，结果存入 70H
地址和进位标志 C。 

 

70H 地址 70H 地址 ＋ （70H+10H）地址 ＋ C 

 

⑤将 X 变址模式标志 T 的内容置 0。 

 

例如：70H 地址的内容为 53H、80H 地址的内容为 21H，在①～⑤的指令执行后 70H 地址的内容变为 74H、
进位标志 C 的内容变为 0。 

（2）減法运算（助记符 SBC） 

SBC 指令在 X 变址模式标志 T 为 1 时将 M（X）的内容减去存储器 M 的内容和进位标志 C 的内容的补码，
结果存入 M（X）和进位标志 C。此时，累加器 A 的内容不变，而状态标志发生变化。其中，M（X）为变址
寄存器Ｘ指向地址的存储器。 

例） 

SEC ① 

SET ② 

LDX #SBDATA ③ 

SBC $10, X ④ 

CLT ⑤ 

 

（说明） 

①将进位标志 C 的内容置 1。 

②将 X 变址模式标志 T 的内容置 1。 

③将#SBDATA（例：$80）装入变址寄存器 X。 

④将存储器的 80H 地址的内容减去 90H 地址（80H+10H）的内容和进位标志 C 的内容的补码，结果存入
80H 地址和进位标志 C。 

 

80H 地址 80H 地址 － （80H+10H）地址 － C 的补码 

 

⑤将 X 变址模式标志 T 的内容置 0。 

 

例如：80H 地址的内容为 53H、90H 地址的内容为 21H，在①～⑤的指令执行后 80H 地址的内容变为 32H、
进位标志 C 的内容变为 1。 

2.2 数据的位判定转移 
（1）置位时转移（助记符 BBS） 

BBS 指令测试累加器 A 或者存储器 M 指定的位 i。如果该位为 1，就转移到指定地址。转移目标地址为相
对地址。如果该位为 0，就执行下一条指令。 

例） 

BBS 0, P0, START 

 

处理 A 

 

START: 

 

处理 B 

 

（说明） 

如果端口 P0 的位 0 为 1，就转移到 START。如果端口 P0 的位 0 为 0，就执行下一条指令。 

 

 
YES 

端口 P00 为“H”电平?  

NO 
 

处理 A 

 

处理 B 

（2）清除时转移（助记符 BBC） 

BBC 指令测试累加器或者存储器 M 指定的位 i。如果该位为 0，就转移到指定地址。转移目标地址为相对
地址。如果该位为 1，就执行下一条指令。 

例） 

RESET: 

 

处理 A 

BBC 7, P1, RESET 

处理 B 

 

 

（说明） 

如果端口 P1 的位 7 为 0，就转移到 RESET。如果端口 P1 的位 7 为 1，就执行下一条指令。 

 

 

 

处理 A 

 

YES 
端口 P17 为“H”电平? 

 
NO 

 
处理 B 

2.3 数据的位处理（置位/复位） 
（1）置位（助记符 SEB） 

SEB 指令将累加器 A 或者存储器 M 指定的位 i 的内容置 1。 

例） 

SEB 7, P0 

 

（说明） 

将端口 P0 的位 7 的内容置 1。 

 

7 6 0 

1 

P0 

 

例如：端口 P0 的内容为 53H，在指令执行后端口 P0 的内容变为 D3H。 

 

（2）位清除（助记符 CLB） 

CLB 指令将累加器 A 或者存储器 M 指定的位 i 的内容清 0。 

例） 

CLB 6, P1 

 

（说明） 

将端口 P1 的位 6 的内容清 0。 

 

7 6 5 0 

0 

P1 

 

例如：端口 P1 的内容为 53H，在指令执行后端口 P1 的内容变为 13H。 

 



740 族 
740 族参考程序集 

RCJ06B0103-0100/Rev.1.00 2005.09 Page 8 of 85 

2.4 数据的循环移位 
（1）左循环（助记符 ROL） 

ROL 指令将累加器 A 或者存储器 M 和进位标志 C 相连，其内容左循环 1 位。累加器 A 或者存储器 M 的
位 0 存放进位标志 C 的内容，进位标志 C 存放累加器 A 或者存储器 M 的位 7 的内容。 

例） 

ROL P0 

 

（说明） 

将端口 P0 和进位标志 C 相连，其内容左循环 1 位。 

 

7 0 

P0 C 

 

例如：端口 P0 的内容为 53H、进位标志 C 的内容为 1，在指令执行后端口 P0 的内容变为 A7H、进位标志
C 的内容变为 0。 

 

（2）移位（助记符 ASL） 

ASL 指令将累加器 A 或者存储器 M 的所有位左移 1 位。此时，累加器 A 或者存储器 M 的位 0 变为 0。另
外，进位标志 C 存放累加器 A 或者存储器 M 的位 7 的内容。 

例） 

ASL P0 

 

（说明） 

将端口 P0 的所有位左移 1 位。 

 

7 0 

C P0 0 

 

例如：端口 P0 的内容为 53H、进位标志 C 的内容为 1，在指令执行后端口 P0 的内容变为 A6H、进位标志
C 的内容变为 0。 

（3）右循环（助记符 ROR） 

ROR 指令将累加器 A 或者存储器 M 和进位标志 C 相连，其内容右循环 1 位。累加器 A 或者存储器 M 的
位 7 存放进位标志 C 的内容，进位标志 C 存放累加器 A 或者存储器 M 的位 0 的内容。 

例） 

ROR P1 

 

（说明） 

将端口 P1 和进位标志 C 相连，其内容右循环 1 位。 

 

7 0 

C P1 

 

例如：端口 P1 的内容为 53H、进位标志 C 的内容为 1，在指令执行后端口 P1 的内容变为 A9H、进位标志
C 的内容变为 1。 

 

（4）右移位（助记符 LSR） 

LSR 指令将累加器 A 或者存储器 M 的所有位右移 1 位。此时，累加器 A 或者存储器 M 的位 7 变为 0。另
外，进位标志 C 存放累加器 A 或者存储器 M 的位 0 的内容 

例） 

LSR P1 

 

（说明） 

将端口 P1 的所有位右移 1 位。 

 

7 0 

0 P1 C 

 

例如：端口 P1 的内容为 53H、进位标志 C 的内容为 1，在指令执行后端口 P1 的内容变为 29H、进位标志
C 的内容变为 1。 

（5）高位和低位的交换（助记符 RRF） 

RRF 指令将存储器 M 的内容右循环 4 位。其结果，存储器 M 的内容的高 4 位和低 4 位进行了交换。 

例） 

RRF P2 

 

（说明） 

将端口 P2 的内容的高 4 位和低 4 位进行交换。 

 

7 4 3 0 

P2 高位  P2 低位 

 

 

例如：端口 P2 的内容为 53H，在指令执行后端口 P2 的内容变为 35H。 

3. 基本处理的程序例子 

3.1 RAM 清除 
（1）概要 

指定区域，清除 RAM。 

 

（2）说明 

从 RAM 地址 SAVEPT 开始清除 LENGTH 字节的 RAM。 

 

RAM 

SAVEPT 

LENGTH 清除 

字节 

（3）流程图 

 
RAM_CLR 

 
(X) ← LENGTH 

 
(A) ← 0 

 
(SAVEPT－1＋(X)) ← (A) 

 
(X) ← (X)－1 

 
NO 

(X) = 0 ? 
 

YES 

RETURN 

 

（4）程序清单 
;******************************************************* 
;  RAM clear routine 
;******************************************************* 
RAM_CLR: 
  LDX #LENGTH  ;RAM length 
  LDA #$00 
RAMCL1: STA SAVEPT-1,X  ;Clear from SAVEPT 
  DEX    ;  - to SAVEPT+LENGTH-1 
  BNE RAMCL1  ;Clear end ? 
  RTS    ;Yes 

3.2 数据传送（RAM） 
（1）概要 

在 RAM 区内传送数据。 

 

（2）说明 

从 RAM 地址 LOADPT 开始将 LENGTH 字节的数据传送到 SAVEPT 开始的连续地址。 

 

RAM 

LOADPT 

LENGTH 
字节 

 
传送 

SAVEPT 

（3）流程图 

 
TR1_DAT 

 
(X) ← LENGTH 

 
(A) ← (LOADPT－1＋(X)) 

 
(SAVEPT－1＋(X)) ← (A) 

 
(X) ← (X)－1 

 
NO 

(X) = 0 ? 
 

YES 

RETURN 

 

（4）程序清单 
;******************************************************* 
;  RAM data transfer routine 
;******************************************************* 
; 
TR1_DAT: 
  LDX #LENGTH  ;RAM length 
TR1_01: LDA LOADPT-1,X  ;Transfer data from LOADPT 
  STA SAVEPT-1,X  ;  - to SAVEPT 
  DEX 
  BNE TR1_01  ;Transfer end ? 
  RTS    ;Yes 

3.3 数据传送（固定 ROM 地址） 
（1）概要 

传送 ROM 区内的数据（固定地址）。 

 

（2）说明 

从 ROM 地址 LOADDT 开始将 LENGTH 字节的数据传送到 SAVEPT 开始的连续 RAM 地址。 

 
RAM 

SAVEPT 
 
 
 
 
 

传送 

ROM 
LOADDT 

LENGTH 
字节 

（3）流程图 

 
TR2_DAT 

 
(X) ← LENGTH 

 
(A) ← (LOADDT－1＋(X)) 

 
(SAVEPT－1＋(X)) ← (A) 

 
(X) ← (X)－1 

 
NO 

(X) = 0 ? 

YES 
RETURN 

 

（4）程序清单 
;******************************************************* 
;  ROM data transfer routine ( address fixed ) 
;******************************************************* 
; 
TR2_DAT: 
  LDX #LENGTH  ;ROM length 
TR2_01: LDA LOADDT-1,X  ;Transfer data from LOADDT 
  STA SAVEPT-1,X  ;  - to SAVEPT 
  DEX 
  BNE TR2_01  ;Transfer end ? 
  RTS    ;Yes 

3.4 数据传送（可变 ROM 地址） 
（1）概要 

传送 ROM 区内的数据（可变地址）。 

 

（2）说明 

从 RAM 地址 ADDRPT＋1 和 ADDRPT 的内容指向的 ROM 地址＋1 开始将 LENGTH 字节的数据传送到
SAVEPT 开始的连续 RAM 地址。 

 

RAM 
SAVEPT 

 
 
 
 
 
 

ADDRPT 数据Ⅰ 
ADDRPT+1 数据Ⅱ 传送 

 
 
 

ROM 
地址 + 1 

 
LENGTH 

字节 

（3）流程图 

 
TR3_DAT 

 
(ADDRPT) ← DAT_ADL 

 
(ADDRPT+1) ← DAT_ADH 

 
(X) ← SAVEPT-1+LENGTH 

 
(Y) ← LENGTH 

 
(T) ← 1 

 
(M(X)) ← ((ADDRPT+1)(ADDRPT)+(Y)) 

 
(X) ← (X)-1 

 
(Y) ← (Y)-1 

 
NO 

(Y) = 0 ? 

YES 
(T) ← 0 

 

RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

（4）程序清单 
;******************************************************* 
;  ROM data transfer routine ( address float ) 
;******************************************************* 
; 
TR3_DAT: 

 LDM #DAT_ADL,ADDRPT 
 LDM #DAT_ADH,ADDRPT+1 

;----------------------------------------------- 
 LDX #SAVEPT-1+LENGTH 
 LDY #LENGTH  ;ROM length 
 SET    ;Transfer data from 

TR3_01: LDA (ADDRPT),Y  ;  - (ADDRPT+1)(ADDRPT)+1 
 DEX    ;  - to SAVEPT 
 DEY    ; 
 BNE TR3_01  ;Transfer end ? 
 CLT    ;Yes 
 RTS 

3.5 数据的重新排序 
（1）概要 

按降序重新排列 RAM 区内的数据。 

 

（2）说明 

从 RAM 地址 SAVEPT 按降序重新排列 LENGTH 字节的数据。 

①首先，将要重新排序的区域最高位地址的存储器内容存入累加器 A。 

 
RAM 

SAVEPT 1EH 
62H 

 
LENGTH 

字节 05H 
10H 
F1H F1H A 

 

②然后，将下一位地址的存储器的内容和累加器 A 的内容进行比较。此时，如果累加器 A 的内容大于等于
存储器的内容，就交换存储器和累加器 A 的内容。 

 
RAM 

SAVEPT 1EH 
62H 

 
LENGTH 

字节 05H 
F1H 10H A 
F1H 

③接着，再将下下一位地址的存储器的内容和累加器 A 的内容进行比较。此时，如果累加器 A 的内容大于
等于存储器的内容，就交换存储器和累加器 A 的内容。 

 
RAM 

SAVEPT 1EH 
62H 

 
LENGTH 

字节 10H 05H A 
F1H 
F1H 

 

 

④按此，重复到要重新排序的区域的最低位地址为止，就能将最小值存入累加器 A，累加器 A 的内容存入
最高位地址的存储器。 

 
RAM 

SAVEPT 1EH 
62H 

 
LENGTH 

字节 10H 
F1H 
05H 05H A 

⑤在最高位地址递减的区域内（从 SAVEPT 的 LENGTH－1 字节）执行①～④。 

 
RAM 

SAVEPT 1EH 
62H 

LENGTH－1 
字节 
 10H 

F1H F1H A 
05H 

 

⑥接着，在最高位地址递减的区域内（从 SAVEPT 的 LENGTH－2 字节）执行①～④。 
 

RAM 
SAVEPT 1EH 

62H 
LENGTH－2 

字节 
 F1H  F1H A 

10H 
05H 

 

⑦按此，将最高位地址逐渐递减且执行①～④，直到最高位地址和最下位地址相等为止，就能按降序重新
排列。 

 

另外，只要在累加器 A 的内容小于存储器的内容时进行②③的存储器和累加器 A 的内容交换，就能按升序
重新排序。 

（3）流程图 

 
SRT_DAT 

 
(X) ← LENGTH 

 
(WORK00) ←(X) 

 
(A) ← (SAVEPT－1＋(X)) 

 
NO 

(A) ≥ (SAVEPT－1＋(X)) ? 
 

YES 

(Y) ← (SAVEPT－1＋(X)) 
 

(SAVEPT－1＋(X)) ← (A) 
 

(A) ← (Y) 
 

(X) ← (X)－1 
 

NO 
(X) = 0 ? 

 
YES 

(X) ← (WORK00) 
 

(SAVEPT－1＋(X)) ← (A) 
 

(X) ← (X)－1 

NO 
(X) = 0 ? 

 
YES 

RETURN 

（4）程序清单 
;******************************************************* 
;  Sort routine 
;******************************************************* 
; 
SRT_DAT: 

 LDX #LENGTH  ;Data length 
SRT_01: STX WORK00 

 LDA SAVEPT-1,X  ;SAVEPT <--> SAVEPT-1+WORK00 
SRT_02: CMP SAVEPT-1,X  ;If use (BCS) 

 BCC SRT_03   ;  - then negative 
 LDY SAVEPT-1,X 
 STA SAVEPT-1,X 
 TYA    ;Minimum data set 

SRT_03: DEX    ;  - to A & Y 
 BNE SRT_02 
 LDX WORK00 
 STA SAVEPT-1,X  ;Minimum data set 
 DEX    ;Next area 
 BNE SRT_01   ;Sort end ? 
 RTS    ;Yes 

3.6 16 位数据加法运算（二进制） 
（1）概要 

将 16 位二进制数据进行加法运算。 

 

（2）说明 

将WORK00＋1、WORK00的内容和WORK01＋1、WORK01的内容相加，结果存入WORK00＋1、WORK00。 

将 X 变址模式标志 T 置 1，不破坏累加器 A 的内容进行运算。 

 

 

低位 WORK00 WORK01 WORK00 

＋ → 
高位 WORK00＋1 WORK01＋1 WORK00＋1 

（3）流程图 

 
ADD_16 

 
(X) ← WORK00 

 
(C)← 0 

 
(T) ← 1 

 
(M(X)) ← (M(X))＋(WORK01)＋(C) 

 
(X) ← (X)＋1 

 
(M(X)) ← (M(X))＋(WORK01＋1)＋(C) 

 
(T) ← 0 

 
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（4）程序清单 
;******************************************************* 
;  16 bits BIN. data addition routine 
;******************************************************* 
; 
ADD_16: 

 LDX #WORK00 
 CLC    ;C flag clear 
 SET    ;T flag set 
 ADC WORK01 
 INX    ;(WORK00+1)(WORK00) + 
 ADC WORK01+1  ;  (WORK01+1)(WORK01) 
 CLT    ;T flag clear 
 RTS 

3.7 16 位数据减法运算（二进制） 
（1）概要 

将 16 位二进制数据进行减法运算。 

 

（2）说明 

将 WORK00＋1、WORK00 的内容减去 WORK01＋1、WORK01 的内容，结果存入 WORK00＋1、WORK00。 

将 X 变址模式标志 T 置 1，不破坏累加器 A 的内容进行运算。 

 

 

低位 WORK00 WORK01 WORK00 

－ → 
高位 WORK00＋1 WORK01＋1 WORK00＋1 

（3）流程图 

 
SUB_16 

 
(X) ← WORK00 

 
(C)← 1 

 
(T) ← 1 

 
(M(X)) ← (M(X))－(WORK01)－(C)的补码 

 
(X) ← (X)＋1 

 
(M(X)) ← (M(X))－(WORK01＋1)－(C)的补码 

 
(T) ← 0 

 
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（4）程序清单 
;******************************************************* 
;  16 bits BIN. data subtraction routine 
;******************************************************* 
; 
SUB_16: 

 LDX #WORK00 
 SEC    ;C flag set 
 SET    ;T flag set 
 SBC WORK01 
 INX    ;(WORK00+1)(WORK00) - 
 SBC WORK01+1  ;  (WORK01+1)(WORK01) 
 CLT    ;T flag clear 
 RTS 

3.8 16 位数据加法运算（BCD） 
（1）概要 

将 16 位 BCD 数据进行加法运算。 

 

（2）说明 

将WORK00＋1、WORK00的内容和WORK01＋1、WORK01的内容相加，结果存入WORK00＋1、WORK00。 

通过将 10 进制模式标志 D 置 1，能使用 10 进制加法运算的 ADC 指令。但是，此时进位标志 C 的确定较
慢，所以请不要紧接在 ADC 指令的后面执行 SEC、CLC、CLD 指令。 

将 X 变址模式标志 T 置 1，不破坏累加器 A 的内容进行运算。 

 

低位 WORK00 WORK01 WORK00 

＋ → 
高位 WORK00＋1 WORK01＋1 WORK00＋1 

（3）流程图 

 
ADD_16D 

 
(X) ← WORK00 

 
(C) ← 0 

 
(T) ← 1 

 
(D) ← 1 

 
(M(X)) ← (M(X))+(WORK01)+(C) 

 
(X) ← (X)+1 

 
(M(X)) ← (M(X))+(WORK01+1)+(C) 

 
(T) ← 0 

 
(D) ← 0 

 
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（4）程序清单 
;******************************************************* 
;  16 bits BCD data addition routine 
;******************************************************* 
; 
ADD_16D: 

 LDX #WORK00 
 CLC    ;C flag reset 
 SET    ;T flag set 
 SED    ;Decimal mode set 
 ADC WORK01 
 INX    ;(WORK00+1)(WORK00) + 
 ADC WORK01+1  ;  (WORK01+1)(WORK01) 
 CLT    ;T flag reset 
 CLD    ;Decimal mode clear 
 RTS 

3.9 16 位数据减法运算（BCD） 
（1）概要 

将 16 位 BCD 数据进行减法运算。 

 

（2）说明 

将 WORK00＋1、WORK00 的内容减去 WORK01＋1、WORK01 的内容，结果存入 WORK00＋1、WORK00。 

通过将 10 进制模式标志 D 置 1，能使用 10 进制加法运算的 SBC 指令。但是，此时进位标志 C 的确定较慢，
所以请不要紧接在 SBC 指令的后面执行 SEC、CLC、CLD 指令。 

将 X 变址模式标志 T 置 1，不破坏累加器 A 的内容进行运算。 

 

低位 WORK00 WORK01 WORK00 

－ → 
高位 WORK00＋1 WORK01＋1 WORK00＋1 

（3）流程图 

 
SUB_16D 

 
(X) ← WORK00 

 
(C) ← 1 

 
(T) ← 1 

 
(D) ← 1 

 
(M(X)) ← (M(X))－(WORK01)－(C)的补码 

 
(X) ← (X)+1 

 
(M(X)) ← (M(X))－(WORK01+1)－(C)的补码 

 
(T) ← 0 

 
(D) ← 0 

 
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（4）程序清单 
;******************************************************* 
;  16 bits BCD data subtraction routine 
;******************************************************* 
; 
SUB_16D: 

 LDX #WORK00 
 SEC    ;C flag set 
 SET    ;T flag set 
 SED    ;Decimal mode set 
 SBC WORK01 
 INX    ;(WORK00+1)(WORK00) - 
 SBC WORK01+1  ;  (WORK01+1)(WORK01) 
 CLT    ;T flag reset 
 CLD    ;Decimal mode reset 
 RTS 

3.10 16 位数据乘法运算（二进制） 
（1）概要 

将 16 位二进制数据进行乘法运算。 

 

（2）说明 

将WORK02＋1、WORK02的内容和WORK01＋1、WORK01的内容相乘，结果存入WORK01＋1、WORK01、
WORK00＋1、WORK00。 

将 X 变址模式标志 T 置 1，不破坏累加器 A 的内容进行运算。 

 

WORK00 最低位 

 
低位 WORK02 WORK01 WORK00+1 

× → 
高位 WORK02+1 WORK01+1  WORK01 

 

WORK01+1 最高位 

（3）流程图 

 
MUL_16 

 
(X) ← WORK00+1 

 
(T) ← 1 

 
(M(X)) ← 00H 

 
(X) ← (X)-1 

 
(M(X)) ← 00H 

 
(Y) ← 16 

1 
(C) ← 0 

 
将(WORK00)和(C)相连，左循环 1 位 

 
将(WORK00+1)和(C)相连，左循环 1 位 

 
将(WORK01)和(C)相连，左循环 1 位 

 
将(WORK01+1)和(C)相连，左循环 1 位 

NO 
(C) = 1 ? 

 
YES 

２ 3 

3 

 
(C) ← 0 

 
( M(X)) ← (M(X))+(WORK02)+(C) 

 
(X) ← (X)+1 

2 
( M(X)) ← (M(X))+(WORK02+1)+(C) 

 
(X) ← (X)-1 

NO 
(C) = 1 ? 

 
YES 

(WORK01) ← (WORK01)+1 

NO 
(WORK01) = 0 ? 

 
YES 

(WORK01+1) ← (WORK01+1)+1 
 

(Y) ← (Y)-1 

NO 
(Y) = 0 ? 1 

 
YES 

(T) ← 0 
 

RETURN 
 

注：M（X）为变址寄存器 X 指向地址的存储器。 

（4）程序清单 
;******************************************************* 
;  16 bits BIN. data multiplication routine 
;******************************************************* 
; 
MUL_16: 

 LDX #WORK00+1  ;Product L addr. set 
 SET    ;T flag set 
 LDA #$00   ;Clear product L 
 DEX 
 LDA #$00 
 LDY #16   ;Bit counter set 

;----------------------------------------------- 
MUL_01: CLC 

 ROL WORK00  ;Rotate product L 
 ROL WORK00+1 
 ROL WORK01  ;Rotate product H 
 ROL WORK01+1 
 BCC MUL_02  ;C flag 1 ? 
 CLC    ;Yes 
 ADC WORK02  ;Multiplicand + product L 
 INX 
 ADC WORK02+1 
 DEX 
 BCC MUL_02  ;Over flow ? 
 INC WORK01  ;Yes 
 BNE MUL_02  ;Over flow ? 
 INC WORK01+1  ;Yes 

;----------------------------------------------- 
MUL_02: DEY 

 BNE MUL_01  ;Multiple end ? 
 CLT    ;Yes 
 RTS 

3.11 16 位数据除法运算（二进制） 
（1）概要 

将 16 位二进制数据进行除法运算。 

 

（2）说明 

将 WORK00＋1、WORK00 的内容除以 WORK02＋1、WORK02 的内容，商存入 WORK00＋1、WORK00，
余数存入 WORK01＋1、WORK01。 

将 X 变址模式标志 T 置 1，不破坏累加器 A 的内容进行运算。 

 

低位 WORK00 WORK02 WORK00 

÷ → 
高位 WORK00+1 WORK02+1  WORK00+1 

 
 
低位 WORK01 

余数 
高位 WORK01+1 

（3）流程图 

 
DIV_16 

 
(X) ← WORK00+1 

 
(Y) ← 16 

 
(T) ← 1 

 
(M(X)) ← 00H 

 
(X) ← (X)+1 

 
(M(X)) ← 00H 

 
(C) ← 0 

1 
将(WORK00)和(C)相连，左循环 1 位 

 
将(WORK00+1)和(C)相连，左循环 1 位 

 
将(WORK01)和(C)相连，左循环 1 位 

 
将(WORK01+1)和(C)相连，左循环 1 位 

NO 
(C) = 0 ? 

 
YES 

2 3 

3 
 

NO 
(M(X)) ≥ (WORK02+1) ? 

2  
YES 

NO 
(M(X)) = (WORK02+1) ? 

 
YES 

(X) ← (X)-1 
 

(M(X))-(WORK02) 
 

(X) ← (X)+1 

NO 
(C) = 1 ? 

 
YES 

(X) ← (X)-1 
 

( M(X)) ← (M(X))－(WORK02)－(C)的补码 
 

(X) ← (X)+1 
 

( M(X)) ← (M(X))－(WORK02+1)－(C)的补码 
 

(C) ← 1 
 

4 

4 

 
(Y) ← (Y)-1 

NO 
(Y) = 0 ? 1 

 
YES 

(T) ← 0 
 

将(WORK00)和(C)相连，左循环 1 位 
 

将(WORK00+1)和(C)相连，左循环 1 位 
 

RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

（4）程序清单 
;******************************************************* 
;  16 bits BIN. data division routine 
;******************************************************* 
; 
DIV_16: 

 LDX #WORK01  ;Surplus addr. set 
 LDY #16   ;Bit counter set 
 SET    ;T flag set 
 LDA #$00   ;Clear surplus 
 INX 
 LDA #$00 
 CLC 

DIV_01: ROL WORK00  ;Rotate quotient 
 ROL WORK00+1 
 ROL WORK01  ;Rotate surplus 
 ROL WORK01+1 
 BCS DIV_02   ;C flag 1 ? 

;----------------------------------------------- 
 CMP WORK02+1  ;No 
 BCC DIV_03  ;Cannot divide ? 
 BNE DIV_02  ;No 
 DEX 
 CMP WORK02 
 INX 
 BCC DIV_03  ;Cannot divide ? 

;----------------------------------------------- 
DIV_02: DEX    ;No 

 SBC WORK02  ;Surplus - divisor 
 INX 
 SBC WORK02+1 
 SEC 

DIV_03: DEY 
 BNE DIV_01  ;Divide  end ? 

;----------------------------------------------- 
 CLT    ;Yes 
 ROL WORK00  ;Rotate quotient 
 ROL WORK00+1 
 RTS 

4. 应用程序例子 

4.1 文件处理（传送） 
（1）概要 

将零页 RAM 部分视为文件存储器，进行数据传送处理。 

 

（2）说明 

从零页 RAM 地址 SAVEPT＋OFFSET 开始将 LENGTH 字节的文件存储器数据传送到 SAVEPT 开始的连续
文件存储器。 

将 X 变址模式标志 T 置位，进行存储器间的传送。另外，指针并非双指针（源/目标）而为单指针＋偏移
量。 

零页 
RAM 

SAVEPT 
 
  

OFFSET 
 

传送 
 
 

LENGTH 
字节 

（3）流程图 

 
TRN_FIL 

 
(A) ← LENGTH 

 
(X) ← SAVEPT 

 
(T) ← 1 

 
(M(X)) ← (OFFSET+(X)) 

 
(X) ← (X)+1 

 
(A) ← (A)-1 

NO 
(A) = 0 ? 

Yes 

(T) ← 0 
 

RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（4）程序清单 
;******************************************************* 
;  File handling (transfer) 
;******************************************************* 
; 
TRN_FIL: 

 LDA #LENGTH  ;File length 
 LDX #SAVEPT 
 SET    ;T flag set 

TRN_01: 
 LDA OFFSET,X  ;Transfer data from 
 INX    ;  - SAVEPT + OFFSET 
 DEC A   ;  - to SAVEPT 
 BNE TRN_01  ;Transfer end ? 
 CLT    ;Yes 
 RTS 

4.2 文件处理（交换） 
（1）概要 

将零页 RAM 部分视为文件存储器，进行数据交换处理。 

 

（2）说明 

将零页RAM地址EXCHPT＋OFFSET开始的LENGTH字节的文件存储器数据和EXCHPT开始的LENGTH
字节的文件存储器数据进行交换。 

将 X 变址模式标志 T 置位，进行存储器间的传送。另外，指针并非双指针（源/目标）而为单指针＋偏移
量。 

零页 
RAM 

EXCHPT 
 
  

OFFSET 
 

交換 
 
 

LENGTH 
字节 

（3）流程图 

 
EXC_FIL 

 
(A) ← LENGTH 

 
(X) ← EXCHPT 

 
(Y) ← (0+(X)) 

 
(T) ← 1 

 
(M(X)) ← (OFFSET+(X)) 

 
(T) ← 0 

 
(OFFSET+(X)) ← (Y) 

 
(X) ← (X)+1 

 
(A) ← (A)-1 

 
NO 

(A) = 0 ? 
 

YES 

RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

（4）程序清单 
;******************************************************* 
;  File handling (exchange) 
;******************************************************* 
; 
EXC_FIL: 

 LDA #LENGTH  ;File length 
 LDX #EXCHPT 

EXC_01: 
 LDY 0,X   ;Exchange data of 
 SET    ;  - EXCHPT + OFFSET 
 LDA OFFSET,X  ;  - with EXCHPT 
 CLT 
 STY OFFSET,X 
 INX 
 DEC A 
 BNE EXC_01  ;Exchange end ? 
 RTS    ;Yes 

4.3 码转换（压缩 BCD→非压缩 BCD） 
（1）概要 

将压缩 BCD 数据转换成非压缩 BCD 数据。 

 

（2）说明 

将零页 RAM 地址 ADDRPT 的压缩 BCD 数据转换成非压缩 BCD，存入 ADDRPT 和 ADDRPT＋1。 

如下所示，压缩 BCD 是将 2 位 BCD 数值压缩成 1 字节的数据。另外，非压缩 BCD 是将压缩 BCD 分成高
4 位为 0 的 2 字节数据。 

 

0 B ADDRPT 

ADDRPT A B → 
0 A ADDRPT+1 

 

压缩 BCD 非压缩 BCD 

（3）流程图 

 
PTOUP 

 
(X) ← ADDRPT 

 
(A) ← (0+(X)) 

 
(T) ← 1 

 
(M(X)) ← (M(X)) ∧ 0FH 

 
(T) ← 0 

 
(A) ← (A) ∧ F0H 

 
(X) ← (X)+1 

 
SWAPA 

 
(0+(X)) ← (A) 

  
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

SWAPA 

 
(T) ← 0 

 
(D) ← 0 

 
(Y) ← 4 

 
将(A)左移 1 位 

 
(A) ← (A)+0+(C) 

 
(Y) ← (Y)-1 

 

NO 
(Y) = 0 ? 

 
YES 

RETURN 

（4）程序清单 
;******************************************************* 
;  Packed BCD -> unpacked BCD 
;******************************************************* 
; 
PTOUP: 

 LDX #ADDRPT 
 LDA 0,X   ;Get packed BCD data 
 SET    ;T flag set 
 AND #0FH   ;Unpacked BCD data L 
 CLT    ;T flag clear 
 AND #0F0H 
 INX 
 JSR SWAPA   ;Swap A 
 STA 0,X   ;Unpacked BCD data H 
 RTS 

;******************************************************* 
;  Swap A register 
;******************************************************* 
; 
SWAPA: 

 CLT    ;T flag clear 
 CLD    ;Decimal mode clear 
 LDY #4 

SWAPA1: 
 ASL A 
 ADC #0 
 DEY 
 BNE SWAPA1 
 RTS 

4.4 码转换（非压缩 BCD→非压缩 BCD） 
（1）概要 

将非压缩 BCD 数据转换成压缩 BCD 数据。 

 

（2）说明 

将零页 RAM 地址 ADDRPT+1 和 ADDRPT 的非压缩 BCD 数据转换成压缩 BCD，存入 ADDRPT。 

如下所示，压缩 BCD 是将 2 位 BCD 数值压缩成 1 字节的数据。另外，非压缩 BCD 是将压缩 BCD 分成高
4 位为 0 的 2 字节数据。 

 
ADDRPT 0 B 

→ A B  ADDRPT 
ADDRPT+1 0 A 

 
非压缩 BCD 压缩 BCD 

（3）流程图 

UPTOP 
 

(X) ← ADDRPT 
 

(X) ← (X)+1 
 

(A) ← (0+(X)) 
 

SWAPA 
 

(X) ← (X)-1 
 

(A) ← (A) ∨ (0+(X)) 
 

(0+(X)) ← (A) 
 

RETURN 

SWAPA 

 
(T) ← 0 

 
(D) ← 0 

 
(Y) ← 4 

 
将(A)左移 1 位 

 
(A) ← (A)+0+(C) 

 
(Y) ← (Y)-1 

 

NO 
(Y) = 0 ? 

 
YES 

RETURN 

（4）程序清单 
;******************************************************* 
;  Unpacked BCD -> packed BCD 
;******************************************************* 
; 
UPTOP: 

 LDX #ADDRPT 
 INX 
 LDA 0,X  ;Get unpacked BCD data H 
 JSR SWAPA  ;Swap A 
 DEX 
 ORA 0,X  ;Packed BCD data 
 STA 0,X 
 RTS 

;******************************************************* 
;  Swap A register 
;******************************************************* 
; 
SWAPA: 

 CLT   ;T flag clear 
 CLD   ;Decimal mode clear 
 LDY #4 

SWAPA1: 
 ASL A 
 ADC #0 
 DEY 
 BNE SWAPA1 
 RTS 

4.5 码转换（BIN→BCD） 
（1）概要 

将 4 字节 BIN 数据转换成 5 字节 BCD 数据。 

 

（2）说明 

将零页 RAM 地址 BINDAT、BINDAT+1、BINDAT+2、BINDAT+3 的４字节 BIN 数据转换成 5 字节的 BCD
数据，存入 BCDDAT、BCDDAT+1、BCDDAT+2、BCDDAT+3、BCDDAT+4。 

 
最高位 BINDAT  BCDDAT 最高位 

 
BINDAT+1 BCDDAT+1 

→ 
BINDAT+2 BCDDAT+2 

  
最低位 BINDAT+3 BCDDAT+3 

 
BCDDAT+4 最低位 

 

 

例）在 BIN 数据为 FFFFFFFFH 的情况下，BCD 数据变为 4294967295。 

 
BINDAT FF16 4210 BCDDAT 

 
FF16 9410 

→ 
FF16 9610 

  
FF16 7210 

 
9510 

（3）流程图 

 
BTOD 

 
(T) ← 1 

 
(D) ← 1 

 
(BCDDAT+4) ← 0 

 
(BCDDAT+3) ← 0 

 
(BCDDAT+2) ← 0 

 
(BCDDAT+1) ← 0 

 
(BCDDAT) ← 0 

 
(Y) ← 32 

 
SFT_BIN 

 
ADD_BCD 

 
(Y) ← (Y)-1 

NO 
(Y) =0 ? 

 
YES 

(D) ← 0 
 

(T) ← 0 
  

RETURN 

SFT_BIN 

 
(C) ← 0 

 
(X) ← 3 

 
将(BINDAT+(X))和(C)相连，左循环 1 位 

 
(X) ← (X)-1 

 

NO 
(X)下溢 ? 

YES 

RETURN 

ADD_BCD 

 
(X) ← BCDDAT+4 

 
(M(X)) ← (M(X))+(0+(X))+(C) 

 
(X) ← (X)-1 

 
(M(X)) ← (M(X))+(0+(X))+(C) 

 
(X) ← (X)-1 

 
(M(X)) ← (M(X))+(0+(X))+(C) 

 
(X) ← (X)-1 

 
(M(X)) ← (M(X))+(0+(X))+(C) 

 
(X) ← (X)-1 

 
(M(X)) ← (M(X))+(0+(X))+(C) 

 
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 



740 族 
740 族参考程序集 

RCJ06B0103-0100/Rev.1.00 2005.09 Page 59 of 85 

（4）程序清单 
;******************************************************* 
;  BIN -> BCD 
;******************************************************* 
; 
BTOD: 

 SET    ;T flag set 
 SED    ;Decimal mode set 
 LDM #0,BCDDAT+4  ;Clear BCD result area 
 LDM #0,BCDDAT+3 
 LDM #0,BCDDAT+2 
 LDM #0,BCDDAT+1 
 LDM #0,BCDDAT 
 LDY #32   ;Yes 

BTOD_01: 
 JSR SFT_BIN  ;Left shift BIN data 
 JSR ADD_BCD  ;2*(BCD)+C -> (BCD) 
 DEY 
 BNE BTOD_01  ;Convert end ? 
 CLD    ;Yes 
 CLT    ;T flag clear 
 RTS 

;******************************************************* 
;  Left shift BIN data 
;******************************************************* 
; 
SFT_BIN: 

 CLC    ;C flag clear 
 LDX #3 

SFT_01: 
 ROL BINDAT,X 
 DEX 
 BPL SFT_01  ;Shift end ? 
 RTS    ;Yes 

;******************************************************* 
;  2*(BCD)+C -> (BCD) 
;******************************************************* 
; 
ADD_BCD: 

 LDX #BCDDAT+4 
 ADC 0,X 
 DEX 
 ADC 0,X 
 DEX 
 ADC 0,X 
 DEX 
 ADC 0,X 
 DEX 
 ADC 0,X 
 RTS 

4.6 码转换（BCD→BIN） 
（1）概要 

将４字节 BCD 数据转换成４字节 BIN 数据。 

 

（2）说明 

将零页 RAM 地址 BCDDAT、BCDDAT+1、BCDDAT+2、BCDDAT+3 的４字节 BCD 数据转换成４字节的
BIN 数据，存入 BINDAT、BINDAT+1、BINDAT+2、BINDAT+3。 

 
最高位 BCDDAT BINDAT 最高位 

 
BCDDAT+1 BINDAT+1 

→ 
BCDDAT+2 BINDAT+2 

 
最低位 BCDDAT+3 BINDAT+3 最低位 

 

 

例）在 BCD 数据为 99999999 的情况下，BIN 数据变为 05F5E0FFH。 

 
BCDDAT 9910  0516 BINDAT 

 
9910 F516 

→ 
9910 E016 

 
9910 FF16 

（3）流程图 
DTOB 

 
(Y) ← 32 

 
(C) ← 0 

 
(X) ← BCDDAT 

 
SFT_DAT 

 

(X) ← BINDAT 
 

SFT_DAT 

 
MOD_BCD 

 
(Y) ← (Y)-1 

 

NO 
(Y) = 0 ？ 
 

YES 

RETURN 

SFT_DAT 
 

(A) ← ４ 
 

将(0+(X))和(C)相连，右循环 1 位 
 

(X) ← (X)+1 
 

(A) ← (A)-1 
 

NO 
(A) = 0 ? 

YES 
RETURN 

MOD_BCD 

 
(X) ← 3 

 
(A) ← (BCDDAT+(X)) 

 

NO 
(A)的位 7 = 1 ? 

 
YES 

(C) ← 1 
 

(A) ← (A)-30H-(C)的补码 
 

NO 
(A)的位 3 = 1 ? 

YES 
(C) ← 1 

 
(A) ← (A)-3-(C)的补码 

 
(BCDDAT+(X)) ←(A) 

 
(X) ← (X)-1 

 

NO 
(X)下溢 ? 

 
YES 

RETURN 

（4）程序清单 
;******************************************************* 
;  BCD -> BIN 
;******************************************************* 
; 
DTOB: 

 LDY #32 
DT0B_01: 

 CLC 
 LDX #BCDDAT  ;Point to BCDDAT 
 JSR SFT_DAT  ;Right shift BCD data 
 LDX #BINDAT  ;Point to BINDAT 
 JSR SFT_DAT  ;Right shift BIN data 
 JSR MOD_BCD  ;Get modified BCD data 
 DEY 
 BNE DT0B_01  ;Shift end ? 
 RTS    ;Yes ! 

;******************************************************* 
;  Right shift data 
;******************************************************* 
; 
SFT_DAT: 

 LDA #4 
SFT_02: 

 ROR 0,X 
 INX 
 DEC A 
 BNE SFT_02 
 RTS 

;******************************************************* 
;  Modify BCD data 
;******************************************************* 
; 
MOD_BCD: 

 LDX #3 
MOD_01: 

 LDA BCDDAT,X 
 BBC 7,A,MOD_02 
 SEC 
 SBC #30H 

MOD_02: 
 BBC 3,A,MOD_03 
 SEC 
 SBC #3 

MOD_03: 
 STA BCDDAT,X 
 DEX 
 BPL MOD_01 
 RTS 

4.7 SGN 函数 
（1）概要 

是累加器 A 的 SGN 函数。 

 

（2）说明 

求累加器 A 的内容的 SGN 值，其结果存入累加器 A。累加器 A 的位 7 为符号位。此时的 SGN 函数如下： 

(A)>0 时 SGN(A)=1 

(A)=0 时 SGN(A)=0 

(A)<0 时 SGN(A)=-1 

（3）流程图 
SGNA 

 

NO 
(A)的位 7 = 0 ? 

 
YES 

YES 
(A)= 0 ?  (A) ← FFH  

 

NO RETURN 

(A) ← 1 
 

RETURN 

 

（4）程序清单 
;******************************************************* 
;  SGN(A) 
;******************************************************* 
; 
SGNA: 

 BBS 7,A,SGN_01 
 CMP #0 
 BEW SGN_02 
 LDA #1 

SGN_02: 
 RTS 

SGN_01: 
 LDA #$FF 
 RTS 

4.8 DCB 12 位浮点四则运算 
（1）概要 

进行 BCD 12 位的浮点四则运算。 

 

（2）说明 

数据格式如下所示。此格式用 7 字节的尾数（1 字节符号位、6 字节数据）和 1 字节的指数表现。尾数的符
号：0 为正、1 为负，尾数的数据为 12 位的有效数据（BCD）。指数只能取 00H～0CH 的数值，如果设定 0DH~FFH
的值，就不保证运算结果。 

 

符号 数据 

 

 

 

 

 

尾数 指数 

 
用此格式表现数的例子如下所示： 

 

1234 0 1 2 3 4 00 
 
 

1.234 0 1 2 3 4 03 
 
 

-15000 1 1 5 0 0 0 00 
 
 

-0.999999999999 1 9 9 9 9 9 9 9 9 9 9 9 9 0C 

 



740 族 
740 族参考程序集 

RCJ06B0103-0100/Rev.1.00 2005.09 Page 67 of 85 

此格式被用于如下所示的 W1、W2、W3 的 3 个文件（零页 RAM 区）。 

 
＋0     ＋1      +2      +3      ＋4     ＋5     ＋6     ＋7 

 

W1 
 
 

W2 
 
 

W3 
 
 

尾数 指数 

 

F16 的位 0 

 

 

●W1：保存被运算数和结果。 

●W2：保存运算数。地址为 W2=W1+8。 

●W3：在进行乘除法运算时，用作工作文件。地址为 W3=W2+8。 

●F16 的位 0：在发生错误时，为 1。 

 

各运算的处理顺序如下所示。加減运算用 W2 的位 3 的值判断，请注意。 

 

·乘法运算 

①给 W1 设定被运算数。 

②给 W2 设定运算数。 

③JSR  MULT 

·除法运算 

①②与乘法运算相同 

③JSR  DIV 

 

·加法运算 

①②与乘法运算相同 

③JSR  ADSB 

 

·減法运算 

①②与乘法运算相同 

③将 W2 的位 3 置 1。 

④JSR  ADSB 

 

在执行这些处理后，能用零页 RAM 地址 F16 的位 0 的数据检查运算结果是否有错。如果（F16 的位 0）=1，
就为错误的运算结果。对于各运算的错误如下所示： 

 

运算名 错误 

加法运算 运算结果超过了 999999999999 

減法运算 运算结果超过了－999999999999 

乘法运算 运算结果超过了±999999999999 

除法运算 ·被 0 除 

·运算结果超过了±999999999999 

（3）程序清单 
;******************************************************* 
;  12 digits BCD number addition,subtraction 
;  ,multiplication,division 
;******************************************************* 
; 
;  RAM ASSIGN 
; 
; 0 1 2 3 4 5 6 7 
; W1: SIGN MSD<------------------------------------LSD INDX 
; W2: SIGN MSD<------------------------------------LSD INDX 
; W3: SIGN MSD<------------------------------------LSD INDX 
; 
;******************************************************* 
;  W1 <- W1/W2 
;******************************************************* 
; 
DIVI: 

 JSR BFITL   ;W3<-W1 copy,W1<-0 
 SBC 8,X   ;D=0,T=1,W1 INDX<-W1 INDX-W2 INDX 
 AND #0FH   ;W1 INDX MSD clear 
 SEB 0,F16 
 BCC DIV1   ;W1 INDX<W2 INDX? 
 CLB 0,F16   ;No 

DIV1: CLT    ;T<-0 
 SED    ;Decimal mode 
 JSR SUB0   ;W1<-W1-W2 
 BCS DIV2   ;Borrow arise? 
 TST W1   ;Yes,then test W1 
 BEQ DIV3 
 DEC W1 

DIV2: CLD    ;Binary mode 
 RRF W3+6 
 CLC 
 LDA W3+6 
 ADC #10H 
 STA W3+6   ;W3 LSD<-W3 LSD+1 
 RRF W3+6 
 BCC DIV1   ;W3 LSD>=15? 

ERROR: SEB 0,F16   ;0  Division error 
 CLD    ;Decimal mode clear 
 RTS 

; 
DIV3: JSR ADD0   ;W1<-W1+W2,add back 

 LDA W3+7 
 CMP #12 
 BCS DIV6 
 INC W3+7 

;      ;W1 MSD<--W1 LSD<--W3 MSD<--W3 LSD 
DIV4: LDY #4   ;4 bits left rotate 
DIV5: LDX #6 

 CLC 
DIV51: ROL W3,X 

 DEX 
 BNE DIV51 
 LDX #6 
 LDA #7 

DIV52: ROL W1,X 
 DEX 
 DEC A 
 BNE DIV52 
 DEY 
 BNE DIV5   ;Rotate end? 
 INX 
 SET    ;T flag set 
 AND #0FH 
 BRA DIV1 

; 
DIV6: LDA W3+1 

 AND #0F0H 
 BNE DIV10 
 BBS 0,F16,DIV7 
 LDA W1+7 
 CMP #12 
 BCS DIV10 

DIV7: INC W1+7 
 BBC 4,W1+7,DIV41 
 CLB 4,W1+7 
 CLB 0,F16 

DIV41: 
 BRA DIV4 

ERROR1: 
 BRA ERROR 

; 
DIV10: BBS 0,F16,ERROR  ;Over flow error 

 LDA W1+7 
 CMP #13 
 BCS ERROR   ;Under flow error 

DIV12: LDY #7   ;W1<-W3 copy 
 LDX #W1 
 SET    ;T<-1 

DIV15: LDA 16,X 
 INX 
 DEY 
 BNE DIV15 

; 
 CLT    ;T<-0 
 JSR INDX   ;"0" condense 
 CLD    ;Decimal mode clear 
 CLC 
 LDA W1 
 ADC W2 
 STA W1   ;Get sign bit 
 CLB 0,F16   ;Normal return 
 RTS 

; 
;******************************************************* 
;  W1 <- W1*W2 
;******************************************************* 
; 
MULT: 

 JSR BFITL   ;W3 <- W1 copy,W1 clear 
 CLC    ;C flag clear 
 ADC 8,X   ;W1_INDX <- W1_INDX+W2_INDX 

 CLB 0,F16 
 BBC 4,W1+7,MULT2 ;W1_INDX <- W_INDX+W2_INDX 
 SEB 0,F16   ;Yes 
 CLB 4,W1+7 

MULT2: 
 CLT    ;T flag clear 

MULT7: 
 LDA W3+6 
 AND #$0F 
 BNE MULT3 
 INC W3+7   ;W1_MSD->W1_LSD->W3_MSD->W3_LSD 

MULT9: 
 LDY #4 

MULT6: 
 LDX #0 
 LDA #7 
 CLC 

MULT4: 
 ROR W1,X 
 INX 
 DEC A 
 BNE MULT4 
 LDX #1 
 LDA #6 

MULT5: 
 ROR W3,X 
 INX 
 DEC A 
 BNE MULT5 
 DEY 
 BNE MULT6   ;4 bits right rotate end ? 
 LDA W3+7 
 CMP #12 
 BCC MULT7 
 BBS 0,F16,MULT8 
 LDX #W1+1   ;W1 "0" test 

MULT81: 
 LDA 0,X 
 BNE MULT8 
 INX 
 CPX #W1+7 
 BNE MULT81 
 BRA DIV12   ;Get result (to division routine) 

MULT8: 
 LDA W1+7 
 BNE MULT10 
 BBC 0,F16,ERROR1 ;Overflow error 
 CLB 0,F16 

MULT10: 
 DEC W1+7 
 LDA W1+7 
 AND #$0F 
 STA W1+7 
 BRA MULT9 

MULT3: 
 DEC W3+6 
 SED 

 JSR ADD0 
 BCC MULT31 
 INC W1 

MULT31: 
 BRA MULT2 

;******************************************************* 
;  W1 <- W1-W2 
;******************************************************* 
; 
SUB0: 

 SEC 
 LDA #6 
 LDX #W1+6 
 SET    ;T flag set 

SUB01: 
 SBC 8,X 
 DEX 
 DEC A 
 BNE SUB01 
 CLT    ;T flag clear 
 RTS 

;******************************************************* 
;  W1 <- W1+W2 
;******************************************************* 
; 
ADD0: 

 CLC 
 LDA #6 
 LDX #W1+6 
 SET    ;T flag set 

ADD01: 
 ADC 8,X 
 DEX 
 DEC A 
 BNE ADD01 
 CLT    ;T flag clear 
 RTS 

;******************************************************* 
;  Following "0" condense,then modify INDX and data 
;******************************************************* 
; 
INDX: 

 TST W1+7   ;Test W1 INDX 
 BEQ INDX1 
 LDA W1+6 
 AND #$0F 
 BNE INDX1   ;Valid data remain ? 
 LDX #W1   ;No 
 JSR SFR4   ;Condense to LSB direction 
 DEC W1+7   ;Modify INDX 
 BRA INDX   ;again 

INDX1: 
 RTS 

;******************************************************* 
;  W1 -> W3 ,and  0 -> W1 
;******************************************************* 
; 

BFITL: 
 CLT 
 LDX #W1 

BFITL1: 
 LDA 0,X 
 STA 16,X 
 LDA #0 
 STA 0,X 
 INX 
 CPX #W1+7 
 BNE BFITL1 
 LDM #0,W3+7  ;W3_INDX initial 
 CLD    ;Decimal mode clear 
 SET    ;T flag set 
 RTS 

;******************************************************* 
;  WN 4 bits left shift 
;******************************************************* 
; 
SFL4: 

 LDA #4 
SFL41: 

 ASL 6,X 
 ROL 5,X 
 ROL 4,X 
 ROL 3,X 
 ROL 2,X 
 ROL 1,X 
 DEC A 
 BNE SFL41 
 RTS 

;******************************************************* 
;  WN 4 bits right shift 
;******************************************************* 
; 
SFR4: 

 LDA #4 
SFR41: 

 LSR 1,X 
 ROR 2,X 
 ROR 3,X 
 ROR 4,X 
 ROR 5,X 
 ROR 6,X 
 DEC A 
 BNE SFR41 
 RTS 

;******************************************************* 
;  Adjust INDX to W1 to W2 
;******************************************************* 
; 
ADIX: 

 LDA W1+7 
 CMP W2+7 
 BEQ ADIX1 
 BCC ADIX23 

ADIX22: 

  LDX #W2 
ADIX2: 

 LDA 1,X 
 AND #$F0 
 BEQ ADIX21 
 CPX #W1 
 BEQ ADIX51 
 LDX #W1 

ADIX5: 
 JSR SFR4 
 DEC 7,X 
 BRA ADIX 

ADIX23: 
 LDX #W1 
 BRA ADIX2 

ADIX51: 
 LDX #W2 
 BRA ADIX5 

ADIX21: 
 JSR SFL4 
 INC 7,X 
 BRA ADIX 

ADIX1: 
 RTS 

;******************************************************* 
;  Execute addition,subtraction 
;******************************************************* 
; 
ADSB: 

 SED    ;Decimal mode set 
 CLT    ;T flag clear 
 JSR ADIX   ;Adjust INDX 

ADSB1: 
 BBS 3,W2,ADSB6 
 BBS 0,W1,ADSB7 
 BBS 0,W2,ADSB10 

ADSB9: 
 JSR ADD0   ;W2 <- W1+W2 
 BCC ADD2   ;C arise ? 
 SEB 0,F16   ;Yes 
 JSR SFR4   ;1 digit right shift 
 DEC W1+7   ;INDX -1 
 BMI ADD3   ;FF ? 
 CLC    ;No 
 LDA W1+1 
 ADC #$10 
 STA W1+1 

ADD2: 
 CLB 0,F16 
 JSR INDX 

ADD3: 
 CLD    ;Decimal mode clear 
 RTS 

ADSB7: 
 BBS 0,W2,ADSB9 

ADSB10: 
 JSR SUB0   ;W1 <- W1-W2 

 BCS SUB2 
 INX 

SUB3: 
 LDA #$99 
 SEC 
 SBC 0,X 
 STA 0,X   ;Get 99-X 
 INX 
 CPX #W1+7 
 BNE SUB3 
 DEX 
 SET 
 LDY #6   ;C=1 

SUB4: 
 ADC #0 
 DEX 
 DEY 
 BNE SUB4 
 CLT    ;Get 100's complement 
 COM W1   ;Get sign bit in bit0 

SUB2: 
 CLB 0,F16 
 JSR INDX 
 CLD    ;Decimal mode clear 
 RTS 

ADSB6: 
 BBS 0,W1,ADSB11 
 BBS 0,W2,ADSB9 
 BRA ADSB10 

ADSB11: 
 BBS 0,W2,ADSB10 
 BRA ADSB9 

5. 替代指令 

5.1 交换累加器 
 

（1）概要 

将累加器的高 4 位和低 4 位进行替换。 

 

（2）流程图 

 
SWAPA 

 
(T) ← 0 

 
(D) ← 0 

 
(Y) ← 4 

 
将(A)左移 1 位 

 
(A) ← (A)+0+(C) 

 
(Y) ← (Y)-1 

 

NO 
(Y) = 0 ? 

 
YES 

RETURN 

（3）程序清单 
;******************************************************* 
;  Swap A register 
;******************************************************* 
; 
SWAPA: 

 CLT   ;T flag clear 
 CLD   ;Decimal mode clear 
 LDY #4 

SWAPA1: 
 ASL A 
 ADC #0 
 DEY 
 BNE SWAPA1 
 RTS 

5.2 颠倒累加器的位顺序 
（1）概要 

将累加器的位数据颠倒排列。使用变址寄存器Ｘ指定的 1 字节工作存储器。 

 

（2）流程图 
 

CTBA 
 

(Y) ← 8 
 

将(A)和(C)相连，左循环 1 位 
 

将(0+(X))和(C)相连，右循环 1 位 
 

(Y) ← (Y)-1 

 

NO 
(Y) = 0 ? 

 
YES 

(A) ← (0+(X)) 
 

RETURN 

 

（3）程序清单 
;******************************************************* 
; Counter bit A register 
;******************************************************* 
; 
CTBA: 
 LDY #8 
CTBA1: 
 ROL A 
 ROR 0,X 
 DEY 
 BNE CTBA1 
 LDA 0,X 
 RTS 

5.3 存储器的置位 
（1）概要 

将变址寄存器 X 指定的存储器的位 0 置“1”。 

 

（2）流程图 

 
SEBM 

 
(T) ← 1 

 
(M(X)) ←(M(X)) ∨ 01H 

 
(T) ← 0 

 
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（3）程序清单 
;******************************************************* 
;  Set bit 0 of M(X) 
;******************************************************* 
; 
SEBM: 

 SET   ;T flag set 
 ORA #1  ;Bit 0 set 
 CLT   ;T flag clear 
 RTS 

5.4 存储器的位清除 
（1）概要 

将变址寄存器 X 指定的存储器的位 0 清“0”。 

 

（2）流程图 
 

CLBM 
 

(T) ← 1 
 

(M(X)) ←(M(X)) ∧ FEH 
 

(T) ← 0 
 

RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（3）程序清单 
;******************************************************* 
;  Clear bit 0 of M(X) 
;******************************************************* 
; 
CLBM: 

 SET   ;T flag set 
 AND #$FE  ;Bit 0 clear 
 CLT   ;T flag clear 
 RTS 

5.5 存储器的位反转 
（1）概要 

将变址寄存器 X 指定的存储器的位 0 进行反转。 

 

（2）流程图 
CMBM 

 
(T) ← 1 

 
(M(X)) ←(M(X))  01H 

 
(T) ← 0 

 
RETURN 

 

注：M（X）为变址寄存器 X 指向地址的存储器。 

 

（3）程序清单 
;******************************************************* 
;  Complement bit 0 of M(X) 
;******************************************************* 
; 
CMBM: 

 SET   ;T flag set 
 EOR #1  ;Bit 0 complement 
 CLT   ;T flag clear 
 RTS 

6. 程序利用的注意事项 
程序利用时，必须注意以下几点。同时请参考 740 族软件手册。 

 

(1) 必须对客户的系统进行充分的评价。 

(2) 本程序通常将 X 变址模式标志 T 和 10 进制模式标志 D 处理为“0”。 请不要在将 X 变址模式标志 T
或 10 进制模式标志 D 置“1”后调用本程序。 

(3) 如果在 10 进制模式时（D=1）执行 ADC 和 SBC 指令，处理器状态寄存器的负标志 N、溢出标志 V 和
零标志 Z 就无效。另外，如果运算结果发生进位，进位标志 C 就被置“1”；如果发生借位，进位标志
C 就被清“0”。 

(4) 在 10 进制模式时（D=1），SEC 指令、CLC 指令或 CLD 指令不能紧接在 ADC 指令或 SBC 指令后使
用。 

 

7. 参考文献 
软件手册：740 族软件手册 

740 Family Software Manual 
 

最新版本请从瑞萨科技网页取得 

http://www.cn.renesas.com 

Cautions 

1. These materials are intended as a reference to assist our customers in the selection of the Renesas 
Technology Corp. product best suited to the customer's application; they do not convey any license 
under any intellectual property rights, or any other rights, belonging to Renesas Technology Corp. or 
a third party.

2. Renesas Technology Corp. assumes no responsibility for any damage, or infringement of any third-
party's rights, originating in the use of any product data, diagrams, charts, programs, algorithms, or 
circuit application examples contained in these materials.

3. All information contained in these materials, including product data, diagrams, charts, programs and 
algorithms represents information on products at the time of publication of these materials, and are 
subject to change by Renesas Technology Corp. without notice due to product improvements or 
other reasons.  It is therefore recommended that customers contact Renesas Technology Corp. or 
an authorized Renesas Technology Corp. product distributor for the latest product information 
before purchasing a product listed herein. 
The information described here may contain technical inaccuracies or typographical errors. 
Renesas Technology Corp. assumes no responsibility for any damage, liability, or other loss rising 
from these inaccuracies or errors. 
Please also pay attention to information published by Renesas Technology Corp. by various means, 
including the Renesas Technology Corp. Semiconductor home page (http://www.renesas.com).

4. When using any or all of the information contained in these materials, including product data, 
diagrams, charts, programs, and algorithms, please be sure to evaluate all information as a total 
system before making a final decision on the applicability of the information and products.  Renesas 
Technology Corp. assumes no responsibility for any damage, liability or other loss resulting from the 
information contained herein.

5. Renesas Technology Corp. semiconductors are not designed or manufactured for use in a device or 
system that is used under circumstances in which human life is potentially at stake.  Please contact 
Renesas Technology Corp. or an authorized Renesas Technology Corp. product distributor when 
considering the use of a product contained herein for any specific purposes, such as apparatus or 
systems for transportation, vehicular, medical, aerospace, nuclear, or undersea repeater use.

6. The prior written approval of Renesas Technology Corp. is necessary to reprint or reproduce in 
whole or in part these materials.

7. If these products or technologies are subject to the Japanese export control restrictions, they must 
be exported under a license from the Japanese government and cannot be imported into a country 
other than the approved destination. 
Any diversion or reexport contrary to the export control laws and regulations of Japan and/or the 
country of destination is prohibited.

8. Please contact Renesas Technology Corp. for further details on these materials or the products 
contained therein.

1. Renesas Technology Corp. puts the maximum effort into making semiconductor products better and 
more reliable, but there is always the possibility that trouble may occur with them. Trouble with 
semiconductors may lead to personal injury, fire or property damage. 
Remember to give due consideration to safety when making your circuit designs, with appropriate 
measures such as (i) placement of substitutive, auxiliary circuits, (ii) use of nonflammable material or 
(iii) prevention against any malfunction or mishap.

Keep safety first in your circuit designs! 

Notes regarding these materials

注意 
本文只是参考译文，前页所载英文版“Cautions”具有正式效力。 

 

1. 

2. 

3. 

 

 

 

http //www.renesas.com

4. 

5. 

6. 

7. 

 

8. 

1. 

